

## ΚΕΦΑΛΑΙΟ 7<sup>ο</sup>

# ΧΡΗΣΙΜΕΣ ΣΥΝΑΡΤΗΣΕΙΣ ΣΤΟ CALC<sup>1</sup>

---

<sup>1</sup> Με τη σημαντική βοήθεια του <https://www.techonthenet.com/excel/index.php> (οι συναρτήσεις δουλεύουν σχεδόν με τον ίδιο τρόπο στο EXCEL).

## Η ΣΥΝΑΡΤΗΣΗ IF

Η συνάρτηση IF συντάσσεται ως

=IF (Test;Then\_value;Otherwise\_value).

Η συνάρτηση αυτή ελέγχει αν ικανοποιείται μια συνθήκη και αποδίδει μια τιμή αν η συνθήκη είναι αληθής (TRUE) και μια άλλη τιμή αν είναι ψευδής (FALSE). Για παράδειγμα, έστω ότι θέλουμε να χαρακτηρίσουμε τα Laptops ως ακριβά ή φθηνά, ανάλογα με την τιμή πώλησής τους.

	A	B	C	
1	<u>AA</u>	Μοντέλο	Τιμή	
2	1	Laptop1	850,00 €	
3	2	Laptop2	770,00 €	
4	3	Laptop3	640,00 €	
5	4	Laptop4	720,00 €	
6	5	Laptop5	495,00 €	
7	6	Laptop6	399,00 €	
8	7	Laptop7	1.050,00 €	
9	8	Laptop8	1.170,00 €	
10	9	Laptop9	960,00 €	
11	10	Laptop10	880,00 €	
12	11	Laptop11	1.250,00 €	
13	12	Laptop12	995,00 €	

Ορίζουμε ως ακριβά αυτά που κοστίζουν >700€ και ως φθηνά αυτά που κοστίζουν <700€. Τότε, για τα δεδομένα στο πλέγμα C2:C13, δίνουμε στο κελί D2 τον τύπο (προσοχή, χρειάζονται τα εισαγωγικά όταν έχουμε κείμενο ως τιμή)

$$=IF(C2>700;"Ακριβό";"Φθηνό")$$

Στη συνέχεια, συμπληρώνουμε τη στήλη με χρήση της αυτόματης συμπλήρωσης. Αν θέλουμε, μπορούμε να βάλουμε IF μέσα στο IF. Για παράδειγμα, έστω ότι κάθε Laptop κατηγοριοποιείται με βάση την τιμή πώλησης ως Ακριβό (>700€), Προσιτό (>450€ και <700€) και Φθηνό (<450€). Τότε, δίνουμε τον εξής τύπο

$$=IF(C2>700;"Ακριβό";IF(C2<450;"Φθηνό";"Προσιτό"))$$

και αποτέλεσμα δίνεται στην παρακάτω εικόνα.

	A	B	C	D	E	F	G
1	<u>AA</u>	Μοντέλο	Τιμή	Χαρακτηρισμός	Πλήθος Ακριβών (>700€)	Χαρακτηρισμός	Πλήθος Ακριβών (>900€)
2	1	Laptop1	850,00 €	Ακριβό	9	Προσιτό	5
3	2	Laptop2	770,00 €	Ακριβό	Πλήθος Φθηνών (<700€)	Προσιτό	Πλήθος Προσιτών (>450€ & <900€)
4	3	Laptop3	640,00 €	Φθηνό	3	Προσιτό	6
5	4	Laptop4	720,00 €	Ακριβό		Προσιτό	Πλήθος Φθηνών (<450€)
6	5	Laptop5	495,00 €	Φθηνό		Προσιτό	1
7	6	Laptop6	399,00 €	Φθηνό		Φθηνό	
8	7	Laptop7	1.050,00 €	Ακριβό		Ακριβό	
9	8	Laptop8	1.170,00 €	Ακριβό		Ακριβό	
10	9	Laptop9	960,00 €	Ακριβό		Ακριβό	
11	10	Laptop10	880,00 €	Ακριβό		Προσιτό	
12	11	Laptop11	1.250,00 €	Ακριβό		Ακριβό	
13	12	Laptop12	995,00 €	Ακριβό		Ακριβό	
14							

## Η ΣΥΝΑΡΤΗΣΗ COUNTIF

Η συνάρτηση αυτή συντάσσεται ως =COUNTIF (range; criteria). Στο range δίνουμε το σύνολο των αριθμητικών δεδομένων και στο criteria δίνουμε τα κριτήρια που θέλουμε να ικανοποιούνται. Για παράδειγμα, έστω ότι θέλουμε για τα δεδομένα που βρίσκονται στο πλέγμα C2:C13 (δείτε προηγούμενη εικόνα), τις τιμές που είναι μεγαλύτερες από 700.

Τότε, θα πρέπει να γράψουμε τον τύπο =COUNTIF (C2 : C13; ">700"). Η COUNTIF κάνει καταμέτρηση των περιπτώσεων (cases) οι οποίες ικανοποιούν τα κριτήρια που βάλουμε (COUNT-IF, δηλ. μέτρησε αν ...). Το αποτέλεσμα δίνεται στην παρακάτω εικόνα:

	A	B	C	D	E	F	G
1	<b>AA</b>	<b>Μοντέλο</b>	<b>Τιμή</b>	<b>Χαρακτηρισμός</b>	<b>Πλήθος Ακριβών (&gt;700€)</b>	<b>Χαρακτηρισμός</b>	<b>Πλήθος Ακριβών (&gt;900€)</b>
2	<b>1</b>	Laptop1	850,00 €	Ακριβό	9	Προσιτό	5
3	<b>2</b>	Laptop2	770,00 €	Ακριβό	<b>Πλήθος Φθηνών (&lt;700€)</b>	Προσιτό	<b>Πλήθος Προσιτών (&gt;450€ &amp; &lt;900€)</b>
4	<b>3</b>	Laptop3	640,00 €	Φθηνό	3	Προσιτό	6
5	<b>4</b>	Laptop4	720,00 €	Ακριβό		Προσιτό	<b>Πλήθος Φθηνών (&lt;450€)</b>
6	<b>5</b>	Laptop5	495,00 €	Φθηνό		Προσιτό	1
7	<b>6</b>	Laptop6	399,00 €	Φθηνό		Φθηνό	
8	<b>7</b>	Laptop7	1.050,00 €	Ακριβό		Ακριβό	
9	<b>8</b>	Laptop8	1.170,00 €	Ακριβό		Ακριβό	
10	<b>9</b>	Laptop9	960,00 €	Ακριβό		Ακριβό	
11	<b>10</b>	Laptop10	880,00 €	Ακριβό		Προσιτό	
12	<b>11</b>	Laptop11	1.250,00 €	Ακριβό		Ακριβό	
13	<b>12</b>	Laptop12	995,00 €	Ακριβό		Ακριβό	
14							

## Η ΣΥΝΑΡΤΗΣΗ COUNTIFS

Η συνάρτηση αυτή συντάσσεται ως

=COUNTIFS (range1; criterial1; range2; criteria2...)

Αποτελεί γενίκευση της COUNTIF αφού μας δίνει τη δυνατότητα να μετρήσουμε τις περιπτώσεις που ικανοποιούν περισσότερα από ένα κριτήρια.

Για παράδειγμα, έστω ότι θέλουμε για τα δεδομένα που βρίσκονται στο πλέγμα C2:C13, τις τιμές που είναι μεγαλύτερες από 450 αλλά και μικρότερες από 900. Θα πρέπει να το δώσουμε ως =COUNTIFS (C2:C13; ">450"; C2:C13; "<900").

	A	B	C	D	E	F	G
1	AA	Μοντέλο	Τιμή	Χαρακτηρισμός	Πλήθος Ακριβών (>700€)	Χαρακτηρισμός	Πλήθος Ακριβών (>900€)
2	1	Laptop1	850,00 €	Ακριβό	9	Προσιτό	5
3	2	Laptop2	770,00 €	Ακριβό	Πλήθος Φθηνών (<700€)	Προσιτό	Πλήθος Προσιτών (>450€ & <900€)
4	3	Laptop3	640,00 €	Φθηνό	3	Προσιτό	6
5	4	Laptop4	720,00 €	Ακριβό		Προσιτό	Πλήθος Φθηνών (<450€)
6	5	Laptop5	495,00 €	Φθηνό		Προσιτό	1
7	6	Laptop6	399,00 €	Φθηνό		Φθηνό	
8	7	Laptop7	1.050,00 €	Ακριβό		Ακριβό	
9	8	Laptop8	1.170,00 €	Ακριβό		Ακριβό	
10	9	Laptop9	960,00 €	Ακριβό		Ακριβό	
11	10	Laptop10	880,00 €	Ακριβό		Προσιτό	
12	11	Laptop11	1.250,00 €	Ακριβό		Ακριβό	
13	12	Laptop12	995,00 €	Ακριβό		Ακριβό	
14							

## Η ΣΥΝΑΡΤΗΣΗ SUMIF

Η συνάρτηση SUMIF συντάσσεται ως

`=SUMIF(range; criteria; sum_range).`

Η συνάρτηση αυτή προσθέτει όλους τους αριθμούς που βρίσκονται στο πλέγμα (range) οι οποίοι ικανοποιούν ένα συγκεκριμένο κριτήριο.

**Παράδειγμα:**

	A	B	C	D	E	F
1	<u>Year</u>	<u>Date</u>	<u>Value</u>	<u>Criteria</u>		37,1▲
2	2000	6/1/2000	11,2	2000		
3	2001	15/12/2001	10,9			
4	2000	19/1/2000	7,2			
5	2002	13/6/2002	4,5			
6	2000	11/10/2000	5,9			
7	2000	9/9/2000	3,5			
8	2004	13/8/2004	12,5			
9	2001	21/3/2001	8,7			
10	2003	29/7/2003	7,4			
11	2000	16/2/2000	9,3			

`=SUMIF (A2 : A11 ; D2 ; C2 : C11 )`

## Η ΣΥΝΑΡΤΗΣΗ SUMIFS

Η συνάρτηση SUMIFS συντάσσεται ως

`SUMIFS (sum_range; range1; criteria1; range2; criteria2; ...)`

Η συνάρτηση αυτή προσθέτει όλους τους αριθμούς που βρίσκονται στο πλέγμα (sum\_range) οι οποίοι ικανοποιούν ένα ή περισσότερα κριτήρια. Στο criteria\_range1 είναι η περιοχή της 1<sup>ης</sup> ομάδας κριτηρίων ενώ το 1<sup>ο</sup> κριτήριο το δίνουμε στο criteria1. Όμοια, στο criteria\_range2 είναι η περιοχή της 2<sup>ης</sup> ομάδας κριτηρίων ενώ το 2<sup>ο</sup> κριτήριο το δίνουμε στο criteria2 κ.ο.κ.

### Παράδειγμα:

	A	B	C	D	E
1	<b>Έτος</b>	<b>Προϊόν</b>	<b>Κόστος</b>		
2	2013	Laptop	200,00 €		400
3	2012	Desktop PC	400,00 €		
4	2012	Printer	100,00 €		
5	2013	Desktop PC	400,00 €		
6	2013	Laptop	200,00 €		
7	2011	Printer	100,00 €		
8	2013	Usb stick	8,00 €		
9	2009	Laptop	200,00 €		

`=SUMIFS (C2:C9;A2:A9;"=2013";B2:B9;"=Laptop")`




## Η ΣΥΝΑΡΤΗΣΗ AVERAGEIF

Η συνάρτηση AVERAGEIF συντάσσεται ως

=AVERAGEIF (range; criteria; average\_range).

Η συνάρτηση αυτή βρίσκει το μέσο όρο των αριθμών που βρίσκονται στο πλέγμα (range) οι οποίοι ικανοποιούν ένα συγκεκριμένο κριτήριο.

**Παράδειγμα:**



	A	B	C	D	E	F
1	<u>Year</u>	<u>Date</u>	<u>Value</u>	<u>Criteria</u>		7,42
2	2000	6/1/2000	11,2	2000		
3	2001	15/12/2001	10,9			
4	2000	19/1/2000	7,2			
5	2002	13/6/2002	4,5			
6	2000	11/10/2000	5,9			
7	2000	9/9/2000	3,5			
8	2004	13/8/2004	12,5			
9	2001	21/3/2001	8,7			
10	2003	29/7/2003	7,4			
11	2000	16/2/2000	9,3			

## Η ΣΥΝΑΡΤΗΣΗ TRIM

Η συνάρτηση TRIM συντάσσεται ως

=TRIM(text).

Η συνάρτηση αυτή επιστρέφει μια τιμή χαρακτήρα (κείμενο, text value) από την οποία έχουν αφαιρεθεί τα κενά (spaces). Μπορεί επίσης να χρησιμοποιηθεί για να αφαιρέσουμε τα μη απαραίτητα κενά μεταξύ των λέξεων σε ένα κείμενο. Παρατηρήστε τι επιστρέφει όταν εφαρμοστεί για το περιεχόμενο στο Κελί A5.

### Παράδειγμα:

	A	B
1	Statistics and Acturial Science	Statistics and Acturial Science
2	83200	83200
3	Statistical Inference	Statistical Inference
4	www.aegean.gr	www.aegean.gr
5	Kar lo vas i	Kar lo vas i
6	Το μάθημα της ΠΛΗΡΟΦΟΡΙΚΗΣ είναι Χειμερινού Εξαμήνου	Το μάθημα της ΠΛΗΡΟΦΟΡΙΚΗΣ είναι Χειμερινού Εξαμήνου
7		

### Τύποι

C1: =TRIM(A1)

C2: = ...


## Η ΣΥΝΑΡΤΗΣΗ LEN

Η συνάρτηση LEN συντάσσεται ως

`=LEN(text)`.

Η συνάρτηση αυτή επιστρέφει το πλήθος των χαρακτήρων ενός κειμένου, λαμβάνοντας υπόψη και τα κενά.

**Παράδειγμα:**



The screenshot shows an Excel spreadsheet with the following data:

	C	D
1	Statistics and Acturial Science	31
2	83200	5
3	Statistical Inference	21
4	www.aegean.gr	13
5	Karlovasi	9
6	Το μάθημα της ΠΛΗΡΟΦΟΡΙΚΗΣ είναι Χειμερινού Εξαμήνου	52
7	Κτίριο Βουρλιώτη, Ισόγειο αριστερά	34

**Σημείωση:** Είτε δώσετε `=LEN("Karlovasi")` είτε `=LEN(Karlovasi)`, το αποτέλεσμα θα είναι το ίδιο (9 χαρακτήρες).

## ΟΙ ΣΥΝΑΡΤΗΣΕΙΣ RIGHT και LEFT

Η συνάρτηση RIGHT συντάσσεται ως

=RIGHT (text; number) .

Η συνάρτηση αυτή επιστρέφει έναν συγκεκριμένο αριθμό χαρακτήρων (number) από το κείμενο (text) ξεκινώντας από τον χαρακτήρα που βρίσκεται δεξιότερα. Με τον ίδιο τρόπο, συντάσσεται και η εντολή LEFT (και προφανώς, η καταμέτρηση ξεκινά από τον χαρακτήρα που βρίσκεται αριστερότερα).

### Παράδειγμα:

	A	B	C	D
1	Statistics and Acturial Science	ence	Stati	ist
2	83200	3200	83200	is
3	Statistical Inference	ence	Stati	stic
4	www.aegean.gr	n.gr	www.a	and Ac
5	Karlovasi	vasi	Karlo	cience
6	Το μάθημα της ΠΛΗΡΟΦΟΡΙΚΗΣ είναι Χειμερινού Εξαμήνου	ήνου	Το μά	
7	Κτίριο Βουρλιώτη, Ισόγειο αριστερά	τερά	Κτίρι	
8				

Τύποι: B1 : =RIGHT (A1 ; 4) , C1 : =LEFT (A1 ; 5) ...

## Η ΣΥΝΑΡΤΗΣΗ MID

Η συνάρτηση MID συντάσσεται ως

$$=MID(\text{text}; \text{start}; \text{number}).$$

Η συνάρτηση αυτή επιστρέφει έναν συγκεκριμένο αριθμό χαρακτήρων από το κείμενο (text) ξεκινώντας από έναν οποιοδήποτε χαρακτήρα (προσδιορίζεται από την τιμή του ορίσματος start\_position).

### Παράδειγμα:

	A	B	C	D
1	<u>Statistics and Acturial Science</u>	ence	Stati	ist
2	83200	3200	83200	is
3	<u>Statistical Inference</u>	ence	Stati	stic
4	<u>www.aegean.gr</u>	n.gr	www.a	and Ac
5	<u>Karlovasi</u>	vasi	Karlo	cience
6	Το μάθημα της <u>ΠΛΗΡΟΦΟΡΙΚΗΣ</u> είναι Χειμερινού Εξαμήνου	ήνου	Το μά	
7	Κτίριο <u>Βουρλιώτη</u> , Ισόγειο αριστερά	τερά	Κτίρι	

## Τύποι

D1: =MID (A1;5;3)

D2: =MID (A1;5;2)

D3: =MID (A1;6;4)

D4: =MID (A1;12;6)

D5: =MID (A1;26;6)

## Η ΣΥΝΑΡΤΗΣΗ CONCATENATE

Η συνάρτηση CONCATENATE συντάσσεται ως

=CONCATENATE (text1 ; text2 ; ...).

Η συνάρτηση αυτή ενώνει 2 ή περισσότερους χαρακτήρες μαζί.

**Παράδειγμα:**

	A	B	C	D	E
1	Text1	Text2	Text3	Text4	Result
2	www	.	aegean.	gr	www.aegean.gr
3	11	8	8	8	11888
4	Σ	A	X	M	ΣΑΧΜ

## Η ΣΥΝΑΡΤΗΣΗ SUBSTITUTE

Η συνάρτηση SUBSTITUTE συντάσσεται ως

`=SUBSTITUTE (text; search_text; new_text; occurrence)`.

Η συνάρτηση αυτή αντικαθιστά ένα σύνολο χαρακτήρων (search\_text) με ένα νέο (new\_text). Μπορούμε να δώσουμε και το πότε θα γίνει η αντικατάσταση (δηλ. σε ποια εμφάνιση του search\_text) χρησιμοποιώντας το όρισμα occurrence (προαιρετικό). Αν δεν υπάρχει αριθμός, τότε κάθε φορά που θα βρίσκει το search\_text θα το αντικαθιστά με το new\_text.

**Παράδειγμα:**

	A	B	C
1	Probability models		Capability models
2	www.aegean.gr		www.a5gean.gr
3			www.aeg5an.gr
4			www.a5g5an.gr



## Τύποι

C2: =SUBSTITUTE (A2;"e";"5";1)

C3: =SUBSTITUTE (A2;"e";"5";2)

C4: =SUBSTITUTE (A2;"e";"5")

## ΟΙ ΣΥΝΑΡΤΗΣΕΙΣ ROUNDUP και ROUNDDOWN

Η συνάρτηση ROUNDUP συντάσσεται ως

=ROUNDUP (number; digits)


και επιστρέφει τον αριθμό στρογγυλοποιημένο προς τα πάνω για δεδομένο αριθμό ψηφίων (digits).

Όμοια, η συνάρτηση ROUNDDOWN συντάσσεται ως

=ROUNDDOWN (number; digits)

και επιστρέφει τον αριθμό στρογγυλοποιημένο προς τα κάτω.

### Παράδειγμα:



The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F	G
1	1823,57		1824	1823			
2	336,2		1823,6	1823,5			
3	3,47		340	330			
4	-3,47		60	50			
5			-23,7	-23,6			
6							

The spreadsheet interface includes a formula bar at the top, a grid of cells, and a sidebar on the right with various icons. The active cell is D6, which is currently empty.

Τύποι:

C1: =ROUNDUP(A1;0)

D1: =ROUNDDOWN(A1;0)

C2: =ROUNDUP(A1;1)

D2: =ROUNDDOWN(A1;1)

C3: =ROUNDUP(A2;-1)

D3: =ROUNDDOWN(A2;-1)

C4: =ROUNDUP(55,1;-1)

D4: =ROUNDDOWN(55,1;-1)

C5: =ROUNDUP(-23,62;1)

D5: =ROUNDDOWN(-23,62;1)

## ΟΙ ΣΥΝΑΡΤΗΣΕΙΣ FLOOR.XCL και CEILING.XCL

Η συνάρτηση FLOOR.XCL συντάσσεται ως

=FLOOR.XCL (number; significance)

και επιστρέφει τον αριθμό στρογγυλοποιημένο προς τα κάτω, **ως προς το κοντινότερο πολλαπλάσιο της τιμής significance**. Όμοια, η συνάρτηση CEILING.XCL συντάσσεται ως


=CEILING.XCL (number; significance)

και επιστρέφει τον αριθμό στρογγυλοποιημένο προς τα πάνω. Με παρόμοιο τρόπο συντάσσονται και οι συναρτήσεις FLOOR και CEILING. Ειδικότερα,

=FLOOR (number; significance; mode)

=CEILING (number; significance; mode)

### Παράδειγμα:



The screenshot shows a spreadsheet with the following data:

	A	B	C	D	E	F	G	H	I	J	K	L
1	1823,57		1824	1823	1823	1823,55	1822	1823	1824	1823,6	1824	1824
2	336,2		1823,6	1823,5	336	336,2	336	336	337	336,2	338	337
3	3,47		340	330	3	3,45	2	3	4	3,5	4	4
4	-3,47				-4	-3,5	-4	-3	-3	-3,45	-2	-4
5												

## Τύποι: Χρησιμοποιώντας FLOOR.XCL, CEILING.XCL

	A	B	C	D	E	F	G
1	1823,57		=ROUNDUP(A1;0)	=ROUNDDOWN(A1;0)	=FLOOR.XCL(A1;1)	=FLOOR.XCL(A1;0,05)	=FLOOR.XCL(A1;2)
2	336,2		=ROUNDUP(A1;1)	=ROUNDDOWN(A1;1)	=FLOOR.XCL(A2;1)	=FLOOR.XCL(A2;0,05)	=FLOOR.XCL(A2;2)
3	3,47		=ROUNDUP(A2;-1)	=ROUNDDOWN(A2;-1)	=FLOOR.XCL(A3;1)	=FLOOR.XCL(A3;0,05)	=FLOOR.XCL(A3;2)
4	-3,47				=FLOOR.XCL(A4;1)	=FLOOR.XCL(A4;0,05)	=FLOOR.XCL(A4;2)
5							
6							
7							

## Τύποι: Χρησιμοποιώντας FLOOR, CEILING

	H	I	J	K	L
2)	=FLOOR(A1;1)	=CEILING.XCL(A1;1)	=CEILING.XCL(A1;0,05)	=CEILING.XCL(A1;2)	=CEILING(A1;1)
2)	=FLOOR(A2;1)	=CEILING.XCL(A2;1)	=CEILING.XCL(A2;0,05)	=CEILING.XCL(A2;2)	=CEILING.XCL(A2;1)
2)	=FLOOR(A3;1)	=CEILING.XCL(A3;1)	=CEILING.XCL(A3;0,05)	=CEILING.XCL(A3;2)	=CEILING.XCL(A3;2)
2)	=FLOOR.XCL(A4;-1)	=CEILING.XCL(A4;1)	=CEILING.XCL(A4;0,05)	=CEILING.XCL(A4;2)	=CEILING.XCL(A4;-1)

## Η ΣΥΝΑΡΤΗΣΗ COUNTA

Η συνάρτηση COUNTA συντάσσεται ως

=COUNTA (value1;value2;...)

και επιστρέφει τον αριθμό των μη-κενών κελιών καθώς επίσης και τον αριθμό των δοθέντων ορισμάτων (arguments).

### Παράδειγμα:

	A	B	C	D	E
1	<b>Επώνυμο</b>	<b>Όνομα</b>	<b>Αντοχή</b>	<b>Ευστοχία</b>	<b>Ομαδικότητα</b>
2	Mills	Patty	A-		B+
3	Green	Dany		A+	B-
4	Leonard	Kawhi	A-		A+
5	Gasol	Paul	A-	C	
6	Aldridge	LaMarcus	B+	B-	
7	Parker	Tony			B+
8	<b>Αριθμός Αθλητών</b>		4	3	4
9					

Τύποι: Στα Κελιά C8, D8, E8 εισάγουμε τους τύπους

=COUNTA (C2 : C7)

=COUNTA (D2 : D7)

=COUNTA (E2 : E7)

και συμπληρώνουμε με χρήση της Αυτόματης Συμπλήρωσης.

## ΟΙ ΣΥΝΑΡΤΗΣΕΙΣ AND και OR

Η συνάρτηση AND συντάσσεται ως

`=AND(Logical value 1; Logical value 2;...)`

και επιστρέφει την τιμή TRUE (ΣΩΣΤΟ) αν όλες οι συνθήκες είναι αληθείς (δηλ. TRUE) ή, επιστρέφει την τιμή FALSE (ΛΑΘΟΣ), αν τουλάχιστον μια από τις συνθήκες είναι ψευδής (FALSE). Όμοια, η συνάρτηση OR συντάσσεται ως

`=OR(Logical value 1; Logical value 2;...)`

και επιστρέφει την τιμή TRUE (ΣΩΣΤΟ) αν τουλάχιστον μια από τις συνθήκες είναι αληθής (δηλ. TRUE) ή, επιστρέφει την τιμή FALSE (ΛΑΘΟΣ), αν όλες οι συνθήκες είναι ψευδείς (δηλ. FALSE).



## Παράδειγμα:

	A	B	C	D
1	27		ΛΑΘΟΣ	ΣΩΣΤΟ
2	<a href="http://www.aegean.gr">www.aegean.gr</a>		ΣΩΣΤΟ	ΛΑΘΟΣ
3			ΣΩΣΤΟ	ΣΩΣΤΟ

=AND (A1>15;A1<38)

Result: TRUE

=AND (A1=30;A2="www.aegeanair.gr")

Result: FALSE

=AND (A1>=25;A1<=32;A2="www.aegean.gr")

Result: TRUE

=OR (A1<15;A1>30)

Result: FALSE

=OR (A1=27;A2="www.aegeanair.gr")

Result: TRUE

=OR (A1>=5;A1=30;A2="www.aegean.gr")

Result: TRUE

C	D
=OR(A1<15;A1>30)	=AND(A1>15;A1<38)
=OR(A1=27;A2="www.aegeanair.gr")	=AND(A1=30;A2="www.aegeanair.gr")
=OR(A1>=5;A1=30;A2="www.aegean.gr")	=AND(A1>=25;A1<=32;A2="www.aegean.gr")

**Παράδειγμα:** Στο αρχείο HomeData.xls (δείτε Berk and Carey (2000)) υπάρχουν δεδομένα σπιτιών (μέγεθος δείγματος 117), για τα οποία έχουν καταγραφεί οι τιμές των παρακάτω μεταβλητών

**Price:** Τιμή πώλησης του σπιτιού

**Square\_foot:** Το μέγεθος του σπιτιού σε τετραγωνικά πόδια (μονάδα μέτρησης εμβαδού, 1 square feet = 0.09290304 square meters)

**Age:** Ηλικία του σπιτιού (σε έτη)

**Features:** Πλήθος αντικειμένων οικιακού εξοπλισμού (π.χ. πλυντήριο πιάτων, ψυγείο, φούρνος μικροκυμάτων, το πολύ 11)

**NE\_Sector:** Βρίσκεται στη νοτιοανατολική πλευρά της πόλης (Yes/No)

**Corner\_Lot:** Το σπίτι βρίσκεται στη γωνία του δρόμου (Yes/No)

**Offer\_Pending:** Εκκρεμεί προσφορά για το σπίτι (Yes/No)

**Annual\_Tax:** Εκτιμώμενο ποσό ετήσιου φόρου ακίνητης περιουσίας

(a) Να βρεθεί ο μέσος ετήσιος φόρος των σπιτιών με ηλικία μικρότερη ή ίση από 10 έτη και μέγεθος > 1600 square feet

Με εφαρμογή της `=AVERAGEIFS(H2:H118;B2:B118;">1600";C2:C118;"<=10")` έπεται ότι το αποτέλεσμα είναι 1075.35

(b) Να βρεθεί ο συνολικός ετήσιος φόρος των σπιτιών με ηλικία μικρότερη από 15 έτη, που δε βρίσκονται στη γωνία του δρόμου και το πλήθος των features είναι 5.

Εδώ έχουμε μια περίπτωση όπου πρέπει να ικανοποιούνται ταυτόχρονα 3 συνθήκες: Age <15, Corner\_Lot = No και Features = 5. Μπορούμε να χρησιμοποιήσουμε έναν συνδυασμό των εντολών IF και AND ώστε να φτιάξουμε μια δείκτρια μεταβλητή με τιμή 1 για τα σπίτια που ικανοποιούν και τις 3 συνθήκες και 0 διαφορετικά.

Αρχικά, δίνουμε στο L2 την εντολή `=IF(AND(C2<15;F2="No";D2=5);1;0)`. Εσωτερικά, υπάρχει η AND με τις συνθήκες C2<15 (για την Age), F2="No" (για την Corner Lot) και D2=5 για την Features. Αν ικανοποιούνται ταυτόχρονα και οι 3, θα επιστρέψει TRUE διαφορετικά θα επιστρέψει FALSE.

Βάζουμε την AND μέσα στην IF και αν ικανοποιούνται και οι 3 συνθήκες, η εφαρμογή (πλέον) της IF θα επιστρέψει την τιμή 1, διαφορετικά θα επιστρέψει την τιμή 0. Με τη βοήθεια της Αυτόματης Συμπλήρωσης, συμπληρώνουμε το array L2:L118 και ουσιαστικά εκτελούμε κάθε φορά (για κάθε σπίτι) έναν έλεγχο αν ικανοποιούνται και οι 3 συνθήκες ή όχι.

Οι τιμές 1 που θα προκύψουν αντιστοιχούν σε σπίτια που ικανοποιούν τις συνθήκες (επιβεβαιώστε ότι είναι συνολικά 4). Πλέον μπορούμε να εφαρμόσουμε την SUMIF() ως εξής:

**=SUMIF(L2:L118;"=1";H2:H118)**

Το αποτέλεσμα είναι 3067.

### Δεύτερη λύση:

Με εφαρμογή της **=SUMIFS(H2:H118,C2:C118,"<15",F2:F118,"=No",D2:D118,"=5")** παίρνουμε το αποτέλεσμα 3067.

(c) Να βρεθεί η συνολική αξία των σπιτιών με ηλικία μεγαλύτερη από 25 έτη, που δε βρίσκονται στη γωνία του δρόμου ή δεν εκρεμμεί προσφορά γι'αυτά.

Θα δουλέψουμε με τον ίδιο τρόπο, συνδυάζοντας αυτή τη φορά τις IF, AND και OR. Εισάγουμε στο M2 την εντολή **=IF(AND(C2>25;OR(F2="No";G2="No")));1;0)**. Αρχικά, με την **OR(F2="No";G2="No")** ελέγχουμε αν το σπίτι δε βρίσκεται στη γωνία του δρόμου ή δεν εκρεμμεί προσφορά γι'αυτό (θέλουμε να ισχύει ένα από τα τα ίδια ή και τα δύο, για να επιστρέψει TRUE) και βάζοντας την OR μέσα στην AND ως

**AND(C2>25;OR(F2="No";G2="No"))**

«απαιτούμε» να ισχύει ταυτόχρονα ότι η ηλικία του σπιτιού είναι >25 έτη και τουλάχιστον μια από τις 2 συνθήκες της OR. Αφού έχουμε συντάξει την AND (σε συνδυασμό με τον OR), την περνάμε μέσα σε μια IF όπου επιστρέφεται η τιμή 1 αν ικανοποιείται η AND, διαφορετικά επιστρέφεται η τιμή 0. Με την αυτόματη συμπλήρωση αντιγράφουμε μέχρι και το M118 και πλέον, με χρήση της SUMIF έχουμε το ζητούμενο αποτέλεσμα (εδώ είναι 1612900)

**=SUMIF(M2:M118;"=1";A2:A118)**

## Η ΣΥΝΑΡΤΗΣΗ VLOOKUP

Η συνάρτηση VLOOKUP συντάσσεται ως

`=VLOOKUP(Search criterion;array;Index;sort order)`

και διεξάγει μια κάθετη αναζήτηση (*a vertical lookup*) ψάχνοντας για μια συγκεκριμένη τιμή (Search criterion) στην 1<sup>η</sup> στήλη του πίνακα (ο οποίος ορίζεται στο array) και επιστρέφει την τιμή που βρίσκεται στην ίδια γραμμή, για τη στήλη που έχουμε υποδείξει στο Index.

- ✓ Συνήθως στο Search criterion χρησιμοποιείται αριθμητική τιμή για λόγους ευκολίας.
- ✓ Επίσης, στο Index, η τιμή που δίνουμε σχετίζεται με τη στήλη που βρίσκεται πιο αριστερά στον πίνακα. Εκείνη η στήλη έχει την τιμή 1 άρα αν, Index=2, τότε θα επιστρέψει την τιμή που βρίσκεται στην ίδια γραμμή που θα συναντήσει το Search criterion στη στήλη 1, για την ακριβώς διπλανή στήλη.
- ✓ Στο sort order δίνουμε είτε TRUE είτε FALSE. Με TRUE, ψάχνει για την πλησιέστερη τιμή που δώσαμε στο value. Αν δώσουμε FALSE, τότε ψάχνει να βρει ακριβώς την

τιμή που δώσαμε στο value. Συνήθως επιλέγεται το FALSE.

### Παράδειγμα:

	A	B	C	D	E	F	G
1	1η Λίστα		2η Λίστα		Κύρια Λίστα		
2	Πωλητής	Εισόδημα	Πωλητής	Όνομα	Πωλητής	Εισόδημα	Όνομα
3	9	13.742,00 €	6	Χρήστος	1	12.442,00 €	Κατερίνα
4	8	14.645,00 €	2	Θανάσης	2	11.025,00 €	Θανάσης
5	11	11.486,00 €	4	Ερασμία	3	16.381,00 €	Μιχάλης
6	6	15.301,00 €	7	Ιωάννα	4	12.336,00 €	Ερασμία
7	4	12.336,00 €	9	Δημήτρης	5	13.446,00 €	Βασίλης
8	5	13.446,00 €	1	Κατερίνα	6	15.301,00 €	Χρήστος
9	7	11.789,00 €	11	Χρήστος	7	11.789,00 €	Ιωάννα
10	2	11.025,00 €	3	Μιχάλης	8	14.645,00 €	Βάσω
11	1	12.442,00 €	5	Βασίλης	9	13.742,00 €	Δημήτρης
12	10	15.249,00 €	8	Βάσω	10	15.249,00 €	Θοδώρα
13	3	16.381,00 €	10	Θοδώρα	11	11.486,00 €	Χρήστος
14	Στο F1 είναι =VLOOKUP (E3 ; \$A\$3 : \$B\$13 ; 2 ; FALSE) Στο G1 είναι =VLOOKUP (E3 ; \$C\$3 : \$D\$13 ; 2 ; FALSE)						
15							
16							
17							

## Η ΣΥΝΑΡΤΗΣΗ HLOOKUP

Η συνάρτηση HLOOKUP συντάσσεται ως

=HLOOKUP (Search criterion; array; Index; sort order)

και διεξάγει μια οριζόντια αναζήτηση (*a horizontal lookup*) ψάχνοντας για μια συγκεκριμένη τιμή (Search criterion) στην 1<sup>η</sup> στήλη του πίνακα (ο οποίος ορίζεται στο array) και επιστρέφει την τιμή που βρίσκεται στην ίδια στήλη, για τη γραμμή που έχουμε υποδείξει στο Index.

Για να γίνει κατανοητή η χρήση της VLOOKUP και της HLOOKUP θα δώσουμε ένα παράδειγμα<sup>2</sup>: Ένα online κατάστημα έχει 3 τύπους παράδοσης για τα προϊόντα του, τον Free με παράδοση εντός 7 ημερών και μηδενικό επιπλέον κόστος, τον Standard με παράδοση εντός 3 ημερών και επιπλέον κόστος 5 ευρώ και τον Premium με παράδοση εντός 1 ημέρας και επιπλέον κόστος 12 ευρώ. Στη διάθεσή μας έχουμε τις παραγγελίες που έκαναν στο κατάστημα 15 πελάτες (κόστος κάθε παραγγελίας, τύπος παράδοσης, ημερομηνία παραγγελίας). Με χρήση της VLOOKUP μπορούμε να έχουμε αυτόματο υπολογισμό του

---

<sup>2</sup> Δείτε επίσης <https://elearn.ellak.gr/mod/page/view.php?id=3016>



ποσού που θα τυπωθεί στην απόδειξη παραγγελίας του πελάτη (και φυσικά, το ποσό που αυτός θα πληρώσει για την παραγγελία του).

Σε ένα κενό φύλλο εργασίας του CALC, εισάγουμε τα παραπάνω στοιχεία

	A	B	C
1	<b>Shipment</b>	<b>Price</b>	<b>Delivery days</b>
2	Free	0,00 €	7
3	Standard	5,00 €	3
4	Premium	12,00 €	1

Τα δεδομένα των παραγγελιών, μαζί με τις επικεφαλίδες, βρίσκονται στο array E1:H16, όπως φαίνεται και στην παρακάτω εικόνα

E	F	G	H	
<u>Customer</u>	<u>Order Total</u>	<u>Shipment</u>	<u>Date of Order</u>	Sh
1	36,00 €	Free	29/10/19	
2	44,00 €	Standard	29/10/19	
3	180,00 €	Standard	29/10/19	
4	93,50 €	Standard	30/10/19	
5	59,00 €	Standard	30/10/19	
6	120,40 €	Premium	30/10/19	
7	100,80 €	Free	30/10/19	
8	93,20 €	Free	30/10/19	
9	45,50 €	Free	30/10/19	
10	87,50 €	Free	31/10/19	
11	39,00 €	Free	31/10/19	
12	73,00 €	Premium	31/10/19	
13	66,50 €	Free	31/10/19	
14	105,00 €	Standard	01/11/19	
15	112,50 €	Premium	01/11/19	

Στα κελιά I1, J1, K1, L1 εισάγουμε τις επικεφαλίδες Shipping Cost, Invoice Total, Delivery Days, Delivery data. Στο κελί I2 εισάγουμε τον τύπο =VLOOKUP(G2;\$A\$2:\$C\$4;2;0) προκειμένου να υπολογίσουμε αυτόματα το επιπλέον κόστος παράδοσης. Για τον 1<sup>ο</sup> πελάτη, η εφαρμογή της VLOOKUP γίνεται ως εξής: Για την τιμή που βρίσκεται στο G2 (εδώ το Free), η συνάρτηση ψάχνει στο A2:C4 (εκεί που βρίσκονται τα στοιχεία για τους διαφορετικούς τύπους παράδοσης, σχετικά με το κόστος και τις ημέρες παράδοσης) και επιστρέφει την τιμή

Πανεπιστήμιο Αιγαίου, Ακαδημαϊκό Έτος 2022-2023

που αντιστοιχί στο Free από τη 2<sup>η</sup> στήλη (γι'αυτό και η τιμή 2 στη σύνταξη της συνάρτησης). Στο τελευταίο όρισμα δίνουμε 0 ώστε αν δε βρει η συνάρτηση αντιστοίχιση, να εμφανίσει μήνυμα σφάλματος. Στη συνέχεια, με τη βοήθεια της Αυτόματης Συμπλήρωσης, αντιγράφουμε τον παραπάνω τύπο μέχρι και το κελί 116. Πλέον, οι τιμές στα J2:J16 προκύπτουν ως το άθροισμα των τιμών στις στήλες Order Total και Shipping Cost.

Μπορούμε να εργαστούμε με τον ίδιο τρόπο για να βρούμε και το πότε θα παραδοθεί η παραγγελία στον πελάτη. Εισάγουμε στο K2 τον τύπο =VLOOKUP(G2;\$A\$2:\$C\$4;3;0), όπου τώρα, για την τιμή που βρίσκεται στο G2, η συνάρτηση ψάχνει στο A2:C4 και επιστρέφει την τιμή που αντιστοιχεί στο Free από τη 3<sup>η</sup> στήλη (γι'αυτό και η τιμή 3 στη σύνταξη της συνάρτησης). Οι τιμές στα L2:L16 προκύπτουν ως το άθροισμα των τιμών στις στήλες Date of Order και Delivery Days.

Η HLOOKUP δουλεύει με ανάλογο τρόπο αλλά σε αναζήτηση γραμμών. Παρακάτω δίνεται ένα απλό παράδειγμα (δεδομένα και συναρτήσεις, πάνω στο φύλλο εργασίας)

	A	B	C	D	E	F	G
1	<b>Φοιτητές</b>	<b>A</b>	<b>B</b>	<b>C</b>	<b>D</b>	<b>E</b>	<b>F</b>
2	Απειροστικός Λογισμός	5	6	5	7	6	7,5
3	Γραμμική Άλγεβρα	6,5	6	5	7,5	5	7
4	Πληροφορική	8	8,5	5	5	9	8,5
5	Στατιστική	5,5	9	7	5,5	6,5	8
6	Εισαγωγή στην Ασφάλιση	9	5	6	9	6	6
7	Μικροοικονομική θεωρία	7	5,5	5	8	6,5	9
8							
9							
10	Βαθμός του D στη Στατιστική	5,5					
11	Βαθμός του A στη Μικροοικονομική Θεωρία	7					
12							
13							

	A	B	C	D	E	F	G
1	Φοιτητές	<u>A</u>	<u>B</u>	<u>C</u>	<u>D</u>	<u>E</u>	<u>F</u>
2	Απειροστικός Λογισμός	5	6	5	7	6	7,5
3	Γραμμική Άλγεβρα	6,5	6	5	7,5	5	7
4	Πληροφορική	8	8,5	5	5	9	8,5
5	Στατιστική	5,5	9	7	5,5	6,5	8
6	Εισαγωγή στην Ασφάλιση	9	5	6	9	6	6
7	Μικροοικονομική θεωρία	7	5,5	5	8	6,5	9
8							
9							
10	Βαθμός του D στη Στατιστική	=HLOOKUP("D";A1:G7;5;0)					
11	Βαθμός του A στη Μικροοικονομική Θεωρία	=HLOOKUP("A";A1:G7;7;0)					
12							

# PIVOT TABLES (ΣΥΓΚΕΝΤΡΩΤΙΚΟΙ ΠΙΝΑΚΕΣ) – ΠΙΝΑΚΕΣ ΔΙΠΛΗΣ ΕΙΣΟΔΟΥ

**Πίνακες Διπλής Εισόδου:** Έστω ότι έχουμε στη διάθεσή μας δεδομένα της μορφής  $(X_i, Y_i)$ ,  $i = 1, 2, \dots, n$ . Οι μεταβλητές  $X, Y$  είναι κατηγορικές με  $I$  κατηγορίες για τη  $X$  (δηλ, η  $X \in \{1, 2, \dots, I\}$ ) και με  $J$  κατηγορίες για τη  $Y$  (δηλ, η  $Y \in \{1, 2, \dots, J\}$ ).

Για παράδειγμα, η  $X$  μπορεί να είναι το φύλο του ατόμου, οπότε η  $X$  λαμβάνει 2 τιμές «ΑΝΔΡΑΣ», «ΓΥΝΑΙΚΑ» (οπότε  $I=2$ ) ενώ η  $Y$  μπορεί να είναι ο δείκτης μάζας σώματος με δυνατές τιμές τις 4 κατηγορίες, δηλ «ΛΙΠΟΒΑΡΗΣ», «ΝΟΡΜΟΒΑΡΗΣ», «ΥΠΕΡΒΑΡΑ», «ΠΑΧΥΣΑΡΚΑ» (οπότε και  $J=4$ ).

Τότε, μπορούμε να καταγράψουμε το πλήθος των μελών του δείγματος τα οποία κατηγοριοποιούνται ως προς το χαρακτηριστικό  $X$  στην  $i$  κατηγορία και ως προς το χαρακτηριστικό  $Y$  στην  $j$  κατηγορία. Αν συμβολίσουμε με  $f_{ij}$  τις συχνότητες αυτών των

μελών του δείγματος (για τα οποία  $X = i, Y = j$ ) τότε  $\sum_{j=1}^J \sum_{i=1}^I f_{ij} = n$  ενώ προκύπτει και ο παραπάνω πίνακας συχνοτήτων (διπλής εισόδου πλέον).

Το πλήθος των κατηγοριών είναι  $I \cdot J$  και καταγράφουμε το σύνολο των ατόμων (από τα συνολικά  $n$ ) τα οποία εμπίπτουν σε κάθε μια από αυτές τις κατηγορίες.

Στο προηγούμενο παράδειγμα, οι κατηγορίες είναι  $2 \cdot 4 = 8$  και συγκεκριμένα είναι

(ΑΝΔΡΑΣ, ΛΙΠΟΒΑΡΗΣ), (ΑΝΔΡΑΣ, ΝΟΡΜΟΒΑΡΗΣ), (ΑΝΔΡΑΣ, ΥΠΕΡΒΑΡΟΣ),

(ΑΝΔΡΑΣ, ΠΑΧΥΣΑΡΚΟΣ), (ΓΥΝΑΙΚΑ, ΛΙΠΟΒΑΡΗΣ), (ΓΥΝΑΙΚΑ, ΝΟΡΜΟΒΑΡΗΣ),

(ΓΥΝΑΙΚΑ, ΥΠΕΡΒΑΡΟΣ), (ΓΥΝΑΙΚΑ, ΠΑΧΥΣΑΡΚΟΣ)

Ουσιαστικά φτιάχνουμε έναν πίνακα συχνοτήτων, μόνο που τώρα είναι προτιμότερο να ταξινομήσουμε τις αντίστοιχες συχνότητες  $f_{ij}$  σε έναν πίνακα διπλής εισόδου, όπως ο επόμενος.

<b><i>X</i></b> \ <b><i>Y</i></b>	<b>1</b>	<b>2</b>	<b>...</b>	<b><i>J</i></b>	<b>Περιθώριο Σύνολο X</b>
<b>1</b>	$f_{11}$	$f_{12}$	<b>...</b>	$f_{1J}$	$f_{1*}$
<b>2</b>	$f_{21}$	$f_{22}$	<b>...</b>	$f_{2J}$	$f_{2*}$
<b>⋮</b>	<b>⋮</b>	<b>⋮</b>	<b>⋮</b>	<b>⋮</b>	<b>⋮</b>
<b><i>I</i></b>	$f_{I1}$	$f_{I2}$	<b>...</b>	$f_{IJ}$	$f_{I*}$
<b>Περιθώριο Σύνολο Y</b>	$f_{*1}$	$f_{*2}$	<b>...</b>	$f_{*J}$	$f_{**} = n$



Επίσης,

$$f_{i*} = \sum_{j=1}^J f_{ij}, i = 1, 2, \dots, I \text{ και } f_{*j} = \sum_{i=1}^I f_{ij}, j = 1, 2, \dots, J.$$

Δεν είναι δύσκολο να μετατρέψουμε τον παραπάνω πίνακα συχνοτήτων σε πίνακα σχετικών συχνοτήτων. Αρκεί να διαιρέσουμε τις τιμές σε όλα τα κελιά  $(i, j)$ ,  $i = 1, 2, \dots, I, j = 1, 2, \dots, J$  αλλά και στα περιθώρια, με το μέγεθος δείγματος  $n$ .

Ένα άλλο παράδειγμα στο οποίο είναι συνήθως απαραίτητη η κατασκευή τέτοιων πινάκων, είναι στην ανάλυση δεδομένων ερωτηματολογίων. Έστω λοιπόν ότι  $X$  = φύλο (με 2 δυνατές κατηγορίες, Α και Β, άρα  $I=2$ ) και  $Y$  είναι μια ερώτηση στην οποία υπάρχουν απαντήσεις από το 1-5 σε μια κλίμακα Likert, π.χ. αξιολογήστε την ποιότητα της παρεχόμενης υπηρεσίας από το 1 έως το 5, με 1="πολύ κακή", 2="κακή", 3="ούτε κακή, ούτε καλή", 4="καλή", 5="πολύ καλή".

Άρα στη διάθεσή μας έχουμε  $n = 30$  άτομα στα οποία έχουμε καταγράψει το φύλο και τι απάντησαν στην αξιολόγηση. Τότε μπορούμε να κατασκευάσουμε τον παρακάτω πίνακα διπλής εισόδου:

<b>X \ Y</b>	<b>1</b>	<b>2</b>	<b>3</b>	<b>4</b>	<b>5</b>	<b>Περιθώριο Σύνολο X</b>
<b>Ανδρας</b>	$f_{11}$	$f_{12}$	$f_{13}$	$f_{14}$	$f_{15}$	$f_{1*}$
<b>Γυναίκα</b>	$f_{21}$	$f_{22}$	$f_{23}$	$f_{24}$	$f_{25}$	$f_{2*}$
<b>Περιθώριο Σύνολο Y</b>	$f_{*1}$	$f_{*2}$	$f_{*2}$	$f_{*2}$	$f_{*2}$	$f_{**} = 30$

Από τον παραπάνω πίνακα μπορούμε να δούμε άμεσα ότι από τα  $n = 30$  άτομα στο δείγμα, τα  $f_{23}$  είναι Γυναίκες που απάντησαν 3 στην ερώτηση αξιολόγησης, ενώ το  $f_{14}$  σημαίνει ότι υπάρχουν μεταξύ των 30 ατόμων του δείγματος  $f_{14}$  το πλήθος τα οποία είναι Άνδρες που απάντησαν 4 στην αξιολόγηση.

Αξίζει επίσης να αναφερθεί ότι οι τιμές στα περιθώρια π.χ. οι τιμές  $f_{1*}$ ,  $f_{2*}$  στο **Περιθώριο Σύνολο X** είναι οι συχνότητες για το χαρακτηριστικό X (δηλ. μας δίνουν την πληροφορία πόσοι Άνδρες και πόσες Γυναίκες είναι στο δείγμα των 30 ατόμων) ενώ αντίστοιχα οι τιμές στο Περιθώριο Σύνολο Y δίνουν την κατανομή των απαντήσεων 1-5 στο δείγμα (πόσοι απάντησαν 1, πόσοι 2, πόσοι 3 κλπ).

Στη συνέχεια θα δούμε πως μπορούμε να κατασκευάσουμε έναν τέτοιο πίνακα με τη βοήθεια του CALC.

Στη διάθεσή μας έχουμε τα δεδομένα στο αρχείο PIVOT\_Tables\_2019.ods στο οποίο έχουμε καταγράψει μέσω ερωτηματολογίου, δημογραφικά στοιχεία. Θέλουμε να κατασκευάσουμε πίνακες συχνοτήτων διπλής εισόδου, δηλαδή να καταγράψουμε το πλήθος των μελών του δείγματος τα οποία ταξινομούνται στις διάφορες κατηγορίες, 2 κατηγορικών μεταβλητών (για τη σχετική διαδικασία, δείτε επίσης<sup>3</sup>).

Αρχικά, θα χρησιμοποιήσουμε τις μεταβλητές Φύλο και Μόρφωση. Επιλέγουμε το σύνολο των δεδομένων στο φύλλο και ακολουθούμε Data / Pivot Table / Current Selection. Επιλέγουμε το Data στο Column Fields και με το πλήκτρο Delete το αφαιρούμε. Στη συνέχεια, με drag-and-drop, τοποθετούμε τη μεταβλητή Φύλο στο Row Fields. Με τον ίδιο τρόπο τοποθετούμε τη μεταβλητή Μόρφωση στο Column Fields. Επίσης, θα τοποθετήσουμε τη μεταβλητή AA στο Data Fields. Αφού την τοποθετήσουμε, πατάμε διπλό αριστερό κλικ πάνω στη μεταβλητή που βάλουμε στο Data Fields και στο παράθυρο διαλόγου επιλέγουμε

---


<sup>3</sup> Δείτε επίσης <https://www.youtube.com/watch?v=0uQfxGNWvic>


Count (Numbers only). Πατάμε OK και ξανά OK και προκύπτει ο παρακάτω πίνακας. Επίσης δίνονται και οι επιλογές στα παράθυρα διαλόγου της διαδικασίας.

	A	B	C	D	E	F
1	Count - AA	Ερ.3 - Μόρφωση				
2	Ερ. 1 - Φύλο	ΑΕΙ/ΤΕΙ	Δευτεροβάθμια	Διδακτορικό	Μεταπτυχιακό	<b>Total Result</b>
3	Άνδρας	6	5	1	2	<b>14</b>
4	Γυναίκα	10	6	1	4	<b>21</b>
5	<b>Total Result</b>	<b>16</b>	<b>11</b>	<b>2</b>	<b>6</b>	<b>35</b>
6						

Παρατηρούμε ότι σε σύνολο 35 ατόμων στο δείγμα, 14 είναι Άνδρες και 21 Γυναίκες. Υπάρχουν επίσης 16 με πτυχίο ΑΕΙ/ΤΕΙ, 11 απόφοιτοι Β'βαθμιας εκπαίδευσης, 2 κάτοχοι διδακτορικού και 6 κάτοχοι μεταπτυχιακού. Επίσης, από τους 16 αποφοίτους ΑΕΙ/ΤΕΙ, οι 10 είναι Γυναίκες ενώ στα 35 άτομα του δείγματος υπάρχουν 2 άνδρες οι οποίοι είναι κάτοχοι μεταπτυχιακού διπλώματος.

Αφού κατασκευάσουμε τον πίνακα, μπορούμε στη συνέχεια να κάνουμε άμεσα και διαγράμματα, όπως π.χ. ραβδογράμματα, είτε ομαδοποιημένα, είτε στοιβαγμένα. Αρκεί να επιλέξουμε τα δεδομένα στο φύλλο εργασίας που περιέχει τον συγκεντρωτικό πίνακα και στη συνέχεια, από το Insert / Chart, θα κατασκευαστεί άμεσα το ομαδοποιημένο ραβδόγραμμα για το Επίπεδο Μόρφωσης, σε Άντρες και Γυναίκες. Αν επιθυμούμε να κατασκευάσουμε το στοιβαγμένο ή το επί τοις 100 (%) στοιβαγμένο ραβδόγραμμα, αρκεί να επιλέξουμε έναν από τους άλλους 2 τύπους ραβδογράμματος.





Μπορούμε να κατασκευάσουμε και τριδιάστατα διαγράμματα.

File Edit View Insert Format Styles Sheet Data Tools Window Help

Pivot Table Layout

**Page Fields:**

**Available Fields:**

- AA
- Ερ.1 – Φύλο
- Ερ.2 – Ηλικία
- Ερ.3 – Μόρφωση
- Ερ.4 – Εξοικείωση με Ηλεκτρονικές
- Ερ.5 – Συχνότητα Ηλεκτρονικών
- Ερ.6 – Κάνετε τις ηλεκτρονικές σ

**Column Fields:**

- Ερ.3 – Μόρφωση

**Row Fields:**

- Ερ.1 – Φύλο

**Data Fields:**

- Sum - AA

Drag the Items into the Desired Position

Options

Source and Destination

Help

OK Cancel

**Data Field**

**Function**

- Average
- Median
- Max
- Min
- Product
- Count (Numbers only)
- StDev (Sample)

Show items without data

Name: AA

Displayed value

Help OK Cancel

	G	H	I
Ερ. 6 – Κάνετε τις ηλεκτρονικές συναλλαγές μέσω			
Smartphone			
PC			
Laptop			
Smartphone			
Tablet			
Smartphone			
Tablet			
Laptop			
Laptop			
Laptop			

25	24 Άνδρας	25-35	Μεταπτυχιακό	Πολλή μεγάλη	5-10
26	25 Γυναίκα	25-35	Μεταπτυχιακό	Πολλή μεγάλη	>=10
27	26 Γυναίκα	36-48	Μεταπτυχιακό	Πολλή μεγάλη	>=10
28	27 Άνδρας	49-65	Δευτεροβάθμια	Μέτρια	5-10
29	28 Άνδρας	25-35	ΑΕΙ/ΤΕΙ	Μέτρια	2-4
30	29 Άνδρας	>65	ΑΕΙ/ΤΕΙ	Μικρή	2-4
31	30 Γυναίκα	25-35	ΑΕΙ/ΤΕΙ	Μεγάλη	5-10
32	31 Γυναίκα	25-35	ΑΕΙ/ΤΕΙ	Πολλή μεγάλη	>=10



Με τον ίδιο τρόπο, μπορούμε να βρούμε το μέσο ύψος συναλλαγών (ή το συνολικό ύψος συναλλαγών) σε Άντρες-Γυναίκες ή μεταξύ αυτών που χρησιμοποιούν Smartphone – PC – Laptop – Tablet στις συναλλαγές τους. Οι αντίστοιχοι πίνακες για συνολικό ύψος συναλλαγών ως προς το Φύλο και ως προς το μέσο συναλλαγών δίνονται παρακάτω.

Ερ. 1 – Φύλο	Sum - Ερ. 7 –	Υψος ηλεκτρονικών συναλλαγών το τελευταίο έτος
Άνδρας	8660	
Γυναίκα	13440	
<b>Total Result</b>	<b>22100</b>	

Ερ. 6 – Κάνει	Sum - Ερ. 7 –	Υψος ηλεκτρονικών συναλλαγών το τελευταίο έτος
Laptop	7500	
PC	1300	
Smartphone	5400	
Tablet	7900	
<b>Total Result</b>	<b>22100</b>	

Σε κάθε συνδυασμό κατηγοριών δίνεται το συνολικό ύψος ηλεκτρονικών συναλλαγών του τελευταίου έτους για τα άτομα που ανήκουν σε αυτό το συνδυασμό

Sum - Ερ. 7 –	Ερ.2 – Ηλικία					
Ερ. 1 – Φύλο	>65	25-35	36-48	49-65		Total Result
Άνδρας	430	5020	1790	1420		<b>8660</b>
Γυναίκα	500	5910	5600	1430		<b>13440</b>
<b>Total Result</b>	<b>930</b>	<b>10930</b>	<b>7390</b>	<b>2850</b>		<b>22100</b>

## ΣΥΝΑΡΤΗΣΕΙΣ ΓΙΑ ΠΡΑΞΕΙΣ ΠΙΝΑΚΩΝ

**Πράξεις Πινάκων:** Να γίνουν οι παρακάτω πράξεις με τη βοήθεια του CALC

(i) Να υπολογιστεί η ορίζουσα του πίνακα

(συνάρτηση **MDETERM**)

$$\mathbf{A} = \begin{bmatrix} 1 & -0,2 & 0,5 & 0,9 \\ -0,2 & 1 & 0,4 & 0,1 \\ 0,5 & 0,4 & 1 & -0,3 \\ 0,9 & 0,1 & -0,3 & 1 \end{bmatrix}$$

**Ορίζουσα πίνακα (Determinant of a matrix):** Έστω ένας τετραγωνικός πίνακας  $\mathbf{A}_{n \times n} = (a_{ij})$ . Ορίζουσα του πίνακα καλούμε την ποσότητα

$$\det(\mathbf{A}) = |\mathbf{A}| = \begin{cases} a_{11}, & n = 1 \\ \sum_{j=1}^n (-1)^{j+1} a_{ij} \det(A_{ij}^*), & n \geq 2 \end{cases}$$

όπου  $i \in \{1, \dots, n\}$  (μπορεί να είναι οποιοδήποτε) και  $A_{ij}^*$  είναι ο  $(n-1) \times (n-1)$  πίνακας που προκύπτει αν διαγράψουμε την  $i$  γραμμή και τη  $j$  στήλη του  $\mathbf{A}$ .

**Λύση:** Εισάγουμε σε ένα πεδίο  $4 \times 4$  έστω αυτό το A1:D4, τις τιμές του παραπάνω πίνακα, όπως αυτές δίνονται. Στο A5 γράφουμε =MDETERM(A1:D4) και πατάμε ENTER για να πάρουμε την τιμή της ορίζουσας.


	A	B	C	D
1	1	-0,2	0,5	0,9
2	-0,2	1	0,4	0,1
3	0,5	0,4	1	-0,3
4	0,9	0,1	-0,3	1
5	-0,7195			

(ii) Να υπολογιστεί ο ανάστροφος του πίνακα

(συνάρτηση TRANSPOSE)

$$A = \begin{bmatrix} 11 & 5 & 0.7 \\ 10 & 5 & 0.8 \\ 9 & 4 & -0.1 \\ 7 & 3 & -0.2 \\ 6 & 5 & 0.3 \\ 9 & 2 & 0.3 \\ 12 & 2 & 0.6 \end{bmatrix}$$

**Λύση:** Εισάγουμε σε ένα πεδίο  $7 \times 3$ , έστω αυτό το A1:C7, τα στοιχεία του πίνακα όπως αυτά δίνονται. Στη συνέχεια, μαρκάρουμε ένα πεδίο διάστασης  $3 \times 7$  και γράφουμε τον τύπο =TRANSPOSE(A1:C7). Πατάμε CTRL + SHIFT + ENTER αφού η TRANSPOSE είναι συνάρτηση πεδίου.



## Ανάστροφος Πίνακα (Transpose of a matrix)

Αν

$$\mathbf{A}_{n \times m} = (x_{ij}) = \begin{pmatrix} a_{11} & a_{12} & \dots & a_{1j} & \dots & a_{1m} \\ a_{21} & a_{22} & \dots & a_{2j} & \dots & a_{2m} \\ \vdots & \vdots & \ddots & \vdots & \ddots & \vdots \\ a_{i1} & a_{i2} & \dots & a_{ij} & \dots & a_{im} \\ \vdots & \vdots & \ddots & \vdots & \ddots & \vdots \\ a_{n1} & a_{n2} & \dots & a_{nj} & \dots & a_{nm} \end{pmatrix}$$

τότε ο ανάστροφος του είναι ο

$$\mathbf{A}'_{m \times n} = \begin{pmatrix} a_{11} & a_{21} & \dots & a_{i1} & \dots & a_{n1} \\ a_{12} & a_{22} & \dots & a_{i2} & \dots & a_{n2} \\ \vdots & \vdots & \ddots & \vdots & \ddots & \vdots \\ a_{1j} & a_{2j} & \dots & a_{ij} & \dots & a_{nj} \\ \vdots & \vdots & \ddots & \vdots & \ddots & \vdots \\ a_{1m} & a_{2m} & \dots & a_{im} & \dots & a_{nm} \end{pmatrix}$$

(iii) Να υπολογιστεί ο αντίστροφος του πίνακα

(συνάρτηση MINVERSE)

$$\mathbf{A} = \begin{bmatrix} 1 & -0,2 & 0,5 & 0,9 \\ -0,2 & 1 & 0,4 & 0,1 \\ 0,5 & 0,4 & 1 & -0,3 \\ 0,9 & 0,1 & -0,3 & 1 \end{bmatrix}$$

**Αντίστροφος πίνακα (Inverse of a matrix):** Έστω  $\mathbf{A}_{n \times n}$  ένας τετραγωνικός πίνακας. Αν υπάρχει πίνακας  $\mathbf{B}_{n \times n}$  τέτοιος ώστε  $\mathbf{A}_{n \times n} \mathbf{B}_{n \times n} = \mathbf{B}_{n \times n} \mathbf{A}_{n \times n} = \mathbf{I}_n$ , τότε ο  $\mathbf{B}$  λέγεται αντίστροφος του  $\mathbf{A}$ .

**Λύση:** Εισάγουμε σε ένα πεδίο 4 × 4 έστω αυτό το A1:D4, τις τιμές του παραπάνω πίνακα, όπως αυτές δίνονται. Στη συνέχεια μαρκάρουμε ένα πεδίο 4 × 4 και γράφουμε =MINVERSE(A1:D4) και πατάμε CTRL + SHIFT + ENTER για να πάρουμε τα στοιχεία του αντίστροφου.

MDETERM				
	A	B	C	D
1	1	-0,2	0,5	0,9
2	-0,2	1	0,4	0,1
3	0,5	0,4	1	-0,3
4	0,9	0,1	-0,3	1
5				
6	=minverse(A1:D4)			
7				
8				
9				
10				

6	-0,99514	-0,82696	1,23280	1,34816
7	-0,82696	0,58374	0,42391	0,81306
8	1,23280	0,42391	-0,14454	-1,19527
9	1,34816	0,81306	-1,19527	-0,65323

(iv) Να προστεθούν (αφαιρεθούν) οι πίνακες

(συνάρτηση + και – ως Array Function)

$$\mathbf{A} = \begin{bmatrix} 2 & 3 & 4 \\ 4 & 9 & 16 \\ 8 & 27 & 64 \\ 16 & 81 & 256 \end{bmatrix}, \mathbf{B} = \begin{bmatrix} 5 & 6 & 7 \\ 25 & 36 & 49 \\ 125 & 216 & 343 \\ 625 & 1296 & 2401 \end{bmatrix}.$$

**Λύση:** Οι πράξεις + και – μπορούν να χρησιμοποιηθούν ως συναρτήσεις πεδίου. Αρχικά εισάγουμε τους πίνακες A και B σε δύο ξεχωριστά πεδία, διάστασης  $4 \times 3$  το καθένα. Στη συνέχεια, επιλέγουμε ένα (κενό) πεδίο  $4 \times 3$  και γράφουμε την εντολή =A1:C4+E1:G4. Πατάμε CTRL+SHIFT+ENTER για να πάρουμε το αποτέλεσμα της πρόσθεσης (το αποτέλεσμα της αφαίρεσης αφήνεται για άσκηση!)

	A	B	C	D	E	F	G
1	2	3	4		5	6	7
2	4	9	16		25	36	49
3	8	27	64		125	216	343
4	16	81	256		625	1296	2401
5							
6	=A1:C4+E1:G4						
7							
8							
9							

	A	B	C	D	E	F	G
1	2	3	4		5	6	7
2	4	9	16		25	36	49
3	8	27	64		125	216	343
4	16	81	256		625	1296	2401
5							
6	7	9	11				
7	29	45	65				
8	133	243	407				
9	641	1377	2657				

Αποτέλεσμα A+B

(v) Να φτιάξετε τον μοναδιαίο πίνακα  $I_4$ . Χρησιμοποιήστε την εντολή `=MUNIT(4)` ως Array Function.

### Μοναδιαίος Πίνακας (Identity Matrix)

$$I_{n \times n} = I_n = \begin{pmatrix} 1 & 0 & \dots & 0 & 0 \\ 0 & 1 & \dots & 0 & 0 \\ \vdots & \vdots & \ddots & \vdots & \vdots \\ 0 & 0 & \dots & 1 & 0 \\ 0 & 0 & \dots & 0 & 1 \end{pmatrix}$$



**Λύση:** Επιλέγουμε ένα πεδίο  $4 \times 4$  και γράφουμε την εντολή =MUNIT(4). Πατάμε CTRL+SHIFT+ENTER και προκύπτει ο μοναδιαίος πίνακας  $4 \times 4$ .

The image shows two screenshots from an Excel spreadsheet. The top screenshot shows the formula bar with 'MDETERM' selected and the formula '=munit(4)' entered. The bottom screenshot shows the formula bar with 'A1:D4' selected and the formula '{=MUNIT(4)}' entered. A purple arrow points from the bottom-left corner of the top screenshot to the bottom-left corner of the bottom screenshot, indicating the transition from entering the formula to the resulting array.

	A	B	C	D
1	=munit(4)			
2				
3				
4				

	A	B	C	D
1	1	0	0	0
2	0	1	0	0
3	0	0	1	0
4	0	0	0	1

(vi) Να λύσετε το γραμμικό σύστημα

$$\begin{aligned}3x + 5y - z &= 1 \\ -2x + 6.5y + 3z &= 0.2 \\ 4x - y + 2.3z &= 5\end{aligned}$$

**Υπόδειξη:** Εξετάστε αν ο πίνακας των συντελεστών

$$\mathbf{A} = \begin{bmatrix} 3 & 5 & -1 \\ -2 & 6.5 & 3 \\ 4 & -1 & 2.3 \end{bmatrix}$$

αντιστρέφεται. Αν συμβαίνει αυτό, τότε το σύστημα έχει μοναδική λύση την

$$\mathbf{x} = \mathbf{A}^{-1}\mathbf{b}$$

όπου  $\mathbf{b}$  είναι το διάνυσμα των σταθερών

$$\mathbf{b} = \begin{bmatrix} 1 \\ 0.2 \\ 5 \end{bmatrix}$$

**Λύση:** Εισάγουμε τον πίνακα  $\mathbf{A}$  σε ένα πεδίο  $3 \times 3$  και το διάνυσμα  $\mathbf{b}$  σε ένα πεδίο  $3 \times 1$ . Αρχικά, θα εξετάσουμε αν αντιστρέφεται ο παραπάνω πίνακας. Θα το κάνουμε υπολογίζοντας την ορίζουσά του, με την εντολή =MDETERM(A1:C3). Η ορίζουσα  $|\mathbf{A}| = 160.85 \neq 0$  και άρα υπάρχει ο  $\mathbf{A}^{-1}$ . Στη συνέχεια, επιλέγουμε ένα πεδίο  $3 \times 1$  και δίνουμε την εντολή =mmult(minverse(A1:C3);E1:E3). Η εντολή αυτή είναι η πράξη  $\mathbf{A}^{-1}\mathbf{b}$ . Πατάμε CTRL + SHIFT + ENTER και θα προκύψει η λύση του παραπάνω συστήματος.

Συνάρτηση για υπολογισμό  
λύσης συστήματος

C5:C7    fx Σ    = {=MMULT(MINVERSE(A1:C3);E1:E3)}

	A	B	C	D	E
1	3	5	-1		1
2	-2	6,5	3		0,2
3	4	-1	2,3		5
4					
5	160,85		0,76686		
6			-0,10084		
7			0,79639		

Υπολογισμός |A|

Εισαγωγή Πίνακα A

Υπολογισμός λύσης  
συστήματος

Εισαγωγή διανύσματος b